

Alkuehtoa vastaavan yksittäisratkaisun etsiminen

Muodostamalla ensin yleinen ratkaisu

Olkoon annettuna differentiaaliyhtälö alkuehtoineen:

```
> diffyht:= x^2*diff(y(x), x, x)-3*x*diff(y(x), x)+4*y(x)=x^3;
```

$$\text{diffyht} := x^2 \left(\frac{d^2}{dx^2} y(x) \right) - 3x \left(\frac{d}{dx} y(x) \right) + 4y(x) = x^3$$

```
> ehto1:= y(2)=3;
```

$$\text{ehto1} := y(2) = 3$$

```
> ehto2:= D(y)(2)=4;
```

$$\text{ehto2} := D(y)(2) = 4$$

Yleinen ratkaisu saadaan **dsolve**-komennolla.

```
> ratkaisu:= dsolve(diffyht, y(x));
```

$$\text{ratkaisu} := y(x) = x^2 _C2 + x^2 \ln(x) _C1 + x^3$$

Määritellään ratkaisu funktioksi $Y(x)$. Tällöin vakiot $_C1$ ja $_C2$ saadaan ratkaistua tekemällä ensin merkkikorvaus $y \rightarrow Y$ edellä annetuille ehdoille ja käyttämällä niitä **solve**-komennon argumentteina.

```
> Y:= unapply(rhs(ratkaisu), x);
```

$$Y := x \rightarrow x^2 _C2 + x^2 \ln(x) _C1 + x^3$$

```
> subs(y=Y, {ehto1, ehto2}):
```

```
C12:= solve(%, {_C1, _C2});
```

$$C12 := \{ _C2 = \frac{3}{2} \ln(2) - \frac{5}{4}, _C1 = \frac{-3}{2} \}$$

Sijoittamalla vakiot yleiseen ratkaisuun saadaan alkuehtoa vastaava yksittäisratkaisu:

```
> y0:= subs(C12, rhs(ratkaisu));
```

$$y0 := x^2 \left(\frac{3}{2} \ln(2) - \frac{5}{4} \right) - \frac{3}{2} x^2 \ln(x) + x^3$$

Tämä toteuttaa differentiaaliyhtälön ja alkuehdot:

```
> subs(y(x)=y0, {diffyht}):
```

```
simplify(%)
```

$$\{ x^3 = x^3 \}$$

```
> subs(x=2, y0);
```

3

```
> subs(x=2, diff(y0,x));
```

4

Ratkaisemalla suoraan dsolve-komennolla

Alkuehdot voidaan yhtälön ohella antaa **dsolve**-komennolle, jolloin yksittäisratkaisu saadaan suoraan:

```
> ratkaisu:= dsolve({diffyht, y(2)=3, D(y)(2)=4}, y(x)):
simplify(%);
```

$$y(x) = \frac{1}{4} x^2 (6 \ln(2) - 5 - 6 \ln(x) + 4x)$$

```
>
```

Linkit

yleinen ja yksittäisratkaisu
alkuehto

ratkaiseminen algebrallisesti Maplella (symalg.mws)

ratkaisun sijoittaminen yhtälöön Maplen keinoin (rtksij.mws)

[SKK & MS 31.05.2001